

SOME EXAMPLES OF THE USE OF COMPUTER ALGEBRA SYSTEMS IN INTRODUCTORY COURSES

Abi Fattahi
Whittier College
Department of Mathematics
Whittier, CA 90605
afattahi@whittier.edu

In this presentation, we will see several examples of mathematics problems solved with the use of a Computer Algebra System. We will note that pattern recognition and discovery-based activities can be utilized to a great extent in solving certain types of problems. Even though the following calculations are done using Maple, other Computer Algebra Systems may be used in a similar way.

In what follows all Maple input are in bold-face and appear left-justified starting with the Maple prompt ">". Most Maple output appear immediately following the input command, and are centered.

1. Find an explicit formula for the exact number of trailing zeros in $n!$.

When given this problem, most students will first have the Computer Algebra System calculate $n!$ for several different values of n , perhaps chosen at random. For example:

```
> 4!;
24
> 10!;
3628800
> 20!;
2432902008176640000
> 40!;
8159152832478977343456112695961158942720000000000
```

If we let $T(n)$ represent the number of trailing zeros in $n!$, it appears that T is a non-decreasing function of n . Further experimentation with different values of n convinces the students that multiples of 5 play an important role in the calculation of $T(n)$. For example, a more astute student, may give the following instruction to Maple:

```
> for n from 0 to 50 by 5 do lprint(n,n!) od;
0 1
5 120
10 3628800
15 1307674368000
20 2432902008176640000
25 15511210043330985984000000
30 265252859812191058636308480000000
35 10333147966386144929666651337523200000000
40 815915283247897734345611269596115894272000000000
45 119622220865480194561963161495657715064383733760000000000
50 30414093201713378043612608166064768844377641568960512000000000000
```


Maple's output is right-justified here so that the trailing zeros can easily be compared. This experimentation has reemphasized the importance of 5, and it is time to look at the problem in a more theoretical way. Now $T(n)$ represents the largest power of 10 that divides $n!$, and there are obviously many more factors of 2 in $n!$ than there are factors of 5. This implies that:

$T(n)$ must be exactly equal to the multiplicity of 5 as a prime factor of $n!$

Now it is clear from the expansion:

$$n! = (1)(2)(3) \dots (n) \quad [1]$$

that each of the factors on the right-hand-side that is a multiple of 5 contributes at least 1 to $T(n)$, and there are exactly $[n/5]$ such factors. On the other hand, each of the factors on the right-hand-side of [1] that is a multiple of 25, contributes one more to $T(n)$, and there are $[n/25]$ such factors. Proceeding in this fashion, we obtain the following expression for the exact value of $T(n)$:

$$T(n) = [n/5] + [n/25] + [n/125] + \dots = \sum_{k=1}^{\infty} \left[\frac{n}{5^k} \right].$$

Note that since $[n/5^k] = 0$ for $5^k > n$, this is really a finite sum.

We can actually use Maple's programming language to define T as a function of n , as given by the above expression:

```
> T := proc(n)
    local k, sum;
    sum := 0;
    for k from 1 while (5^k <= n ) do
        sum := sum + trunc(n/5^k);
    od;
    sum;
end;
```

Now by invoking this function, we can immediately calculate $T(n)$ for any n . For example:

```
> T(563);
```

138

2. Use the graphing capability of a CAS to sketch the outline of a car. To accomplish this, make a rough sketch of a car on the paper, select appropriate curves to represent various sections of the car, and define functions whose graphs will coincide with these curves.

The students are told that outlines consisting of only straight lines will not be sufficient, and that they should make their sketch as realistic as possible.

Having made a preliminary sketch on the paper, one may define two functions whose graphs, when drawn on the same axes, produce the top, and the bottom outlines of the car. These are defined below as step functions:


```

> f := proc(x)
    if x < 0 then x
    elif x < 2 then x^3/8
    elif x < 4 then -1 + sqrt(6*x - x^2 - 4)
    elif x < 5 then -x + 5
    elif x <= 8 then -1 + sqrt(10*x - x^2 - 16)/3
    else -1
    fi
end;

> g := proc(x)
    if x < 3/8 then -1
    elif x < 13/8 then -1+sqrt(2*x-x^2-39/64)
    elif x < 43/8 then -1
    elif x < 53/8 then -1+sqrt(12*x-x^2-2279/64)
    else -1
    fi
end;

```

```

> Top := plot(f, -1 ..8, -2.4 .. 3, scaling = constrained):

```

```

> Bottom := plot(g, -1 ..8, -2.4 .. 3, scaling = constrained):

```

Now define the wheels as the graph of two circles of appropriate size in parametric form:

```

> Wheels := plot([ [1 + 1/2*cos(t) , -1 + 1/2*sin(t), t = 0 .. 2*Pi],
    [6 + 1/2*cos(t) , -1 + 1/2*sin(t), t = 0 .. 2*Pi]],
    thickness = 3, numpoints = 200):


```

Finally the display command will sketch all the graphs on the same axes:

```


> with(plots):
> display({Top, Bottom, Wheels});

```



We can ask Maple to remove the axes:


```
> display({Top, Bottom, Wheels}, axes = none);
```



3. Find an explicit formula for $F_n(x) = \int x^n e^x dx$.

For a specific value of n , a CAS such as Maple can give the exact antiderivative:

```
> int(x^2 * exp(x) , x);
```

$$x^2 e^x - 2x e^x + 2 e^x$$

```
> int(x^5 * exp(x) , x);
```

$$x^5 e^x - 5x^4 e^x + 20x^3 e^x - 60x^2 e^x + 120x e^x - 120 e^x$$

```
> int(x^10 * exp(x) , x);
```

$$x^{10} e^x - 10 x^9 e^x + 90 x^8 e^x - 720 x^7 e^x + 5040 x^6 e^x - 30240 x^5 e^x + 151200 x^4 e^x - 604800 x^3 e^x + 1814400 x^2 e^x - 3628800 x e^x + 3628800 e^x$$

What about the general case? Let's see if Maple can do it:

```
> int(x^n * exp(x) , x);
```

$$\int x^n e^x dx$$

As you can see Maple is not able to evaluate the integral in the general case. A simple application of integration by parts will show that the function F_n satisfies the reduction equation:

$$F_n(x) = x^n e^x - n F_{n-1}(x) \quad [2]$$

It is basically this equation that Maple uses to evaluate the integral for the specific cases of n . On the other hand, a careful look at the special cases above reveals a very interesting property of $F_n(x)$:

It appears that $F_n(x)$ can be expressed as a product of e^x with an n^{th} degree monic polynomial in x , where each term is the negative of the derivative of the previous term.

If this were true in general, then we would have the explicit formula we were seeking in the form of:

$$G_n(x) = e^x \sum_{k=0}^n (-1)^k (x^n)^{(k)}. \quad [3]$$

Having discovered this expression through pattern recognition, it is now easy to see that in fact the function defined by [3] is an anti-derivative for $x^n e^x$:

$$\begin{aligned} G_n'(x) &= e^x \left[\sum_{k=0}^n (-1)^k (x^n)^{(k)} + \sum_{k=0}^n (-1)^k (x^n)^{(k+1)} \right] \\ &= e^x \left[\sum_{k=0}^n (-1)^k (x^n)^{(k)} + \sum_{k=0}^{n-1} (-1)^k (x^n)^{(k+1)} \right] \\ &= e^x \left[x^n + \sum_{k=1}^n (-1)^k (x^n)^{(k)} + \sum_{k=0}^{n-1} (-1)^k (x^n)^{(k+1)} \right]. \end{aligned}$$

Since the last two summations cancel each other, the right-hand-side reduces to $e^x x^n$, proving $G_n'(x) = F_n'(x)$.